


How to start 2 programs simultaneously in windows command prompt



Hi i am using windows 7 64bit

Here is the code snippet i am using to start

```
@echo off
call "C:\Program Files (x86)\LOLReplay\LOLRecorder.exe"
call "G:\League of Legends\lol.launcher.exe"
exit
```

But unless i close LOLRecorder.exe it wont start my lol.launcher.exe.... basically i want both running and the cmd prompt exit after they start. Whats wrong here? I checked out another stackoverflow answer [Here](#) but it refers to the same method i am using.

EDIT:

With the start command it just starts 2 terminal windows and nothing starts!

```
@echo off
start "C:\Program Files (x86)\LOLReplay\LOLRecorder.exe"
start "G:\League of Legends\lol.launcher.exe"
exit
```

[windows-7](#) [command-line](#) [cmd](#) [command-prompt](#) [windows-console](#)

edited May 23 '17 at 12:29



Community ♦
1 1

asked Jun 27 '11 at 1:00



[footy](#)
2,627 8 36 86

4 Answers

With the start command it just starts 2 terminal windows and nothing starts!

The problem is the quotes (which are unfortunately required, due to the spaces in the paths). The `start` command doesn't seem to like them.

You can work around this by using the short DOS names for all the directories (and remove quotes), or by specifying the directory separately and quoting it (which the `start` command seems to be able to deal with).

Try this:

```
@echo off
start /d "C:\Program Files (x86)\LOLReplay" LOLRecorder.exe
start /d "G:\League of Legends" lol.launcher.exe
```

Or, if your batch files become more complicated in the future, or your program names have spaces in them, this:

```
@ECHO OFF

CALL :MainScript
GOTO :EOF


:MainScript
CALL :RunProgramAsync "C:\Program Files (x86)\LOLReplay\LOLRecorder.exe"
CALL :RunProgramAsync "G:\League of Legends\lol.launcher.exe"
GOTO :EOF

:RunProgramAsync
REM ~s1 expands the variable to contain short DOS names only
start %~s1
GOTO :EOF
```

answered Jun 27 '11 at 1:55



[Merlyn Morgan-Graham](#)
45.1k 9 96 152



start requires parameters for window title. Try: start "Lolrecorder" "C:\Program Files (x86)\LOLReplay\LOLRecorder.exe" start "Lol-Launcher" "G:\League of Legends\lol.launcher.exe"

This will give the cmd-windows started by start the title of "Lolrecorder" and "Lol-Launcher"

answered Jun 19 '13 at 4:15



`call` is for batch files only, and it waits for the callee to return. You should use the `start` command to start programs backgrounded. As an added bonus you can specify a priority for the process. If you need to run something as another user, use `runas`.

answered Jun 27 '11 at 1:14

**Kerrek SB**

337k 57 623 859

i have edited my question. Please check it out – [footy](#) Jun 27 '11 at 1:22

@footy: Do either of the two commands individually (without `start`) start the respective program correctly, or are you maybe missing some essential command line options? (Oh, no need to say "exit" I think.) – [Kerrek SB](#) Jun 27 '11 at 1:28

It works individually. There are no cmd line args i have to give to start these programs. – [footy](#) Jun 27 '11 at 1:35

Specify a title and the `/c` switch to tell the STARTed window to go away after its command finishes.

```
start "recorder" /c "C:\Program Files (x86)\LOLReplay\LOLRecorder.exe"
start "LOL" /c "G:\League of Legends\lol.launcher.exe"
```

This [reference](#) has so far answered *almost* every question I've ever had about CMD.

answered Mar 21 '14 at 14:53

**gws**

70 11